# Maintainer: Andrea Dalle Vacche

_realname=Date-Calc
pkgname=perl-${_realname}
pkgver=6.4
pkgrel=1
pkgdesc="Date::Calc - Gregorian calendar date calculations"
arch=('any')
license=('GPL' 'PerlArtistic')
url="https://metacpan.org/release/Date-Calc"
depends=('perl')
groups=('perl-modules')
options=('!emptydirs')
source=(https://www.cpan.org/authors/id/S/ST/STBEY/${_realname}-${pkgver}.tar.gz)
sha256sums=('7ce137b2e797b7c0901f3adf1a05a19343356cd1f04676aa1c56a9f624f859ad')

build() {
  cd ${_realname}-${pkgver}
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd ${_realname}-${pkgver}
  make test
}

package() {
  cd ${_realname}-${pkgver}
  make DESTDIR="${pkgdir}" install
}
